

Einschwingvorgänge or How to turn on an oscillation (part II).

Coupled Oscillations

1 Modulated oscillations on the circle

1.1 The symmetric oscillator and the oscillation on the circle

So far we have considered the equation of a harmonic oscillator based on the physical system (a unit mass with a spring) and interpreting the variable x as the position of the oscillator and v as the velocity in that case the equations were:

$$\dot{x} = v \quad (1a)$$

$$\dot{v} = -\omega^2 x \quad (1b)$$

with angular frequency $\omega = \sqrt{k}$, which gives rise in general to orbits that form concentric ellipses in phase space.¹ The higher the angular velocity ω , the more elongated the ellipse will be in the vertical direction. However, we can symmetrize these equations by making a change of coordinates ($x = \omega x$, $y = v$) and then the equations become:

$$\dot{x} = \omega y \quad (2a)$$

$$\dot{y} = -\omega x \quad (2b)$$

and we can interpret this new x variable as position scaled by the frequency of the oscillator (with the same physical units of a velocity) and of course y is just the velocity of the oscillator. Now the trajectories describe perfect circles with angular velocity ω and the radius determined by the initial condition $r = \sqrt{x_0^2 + y_0^2}$. This can be quickly verified by making an additional change of coordinates now to polar coordinates. The chain rule gives us the possibility to transform the time derivatives in x and y to the derivatives in the polar coordinates for the radius r and for the angle θ in the following way:

$$\dot{r} = \frac{1}{r}(x\dot{x} + y\dot{y}) \quad (3a)$$

$$\dot{\omega} = -\frac{1}{1+y^2/x^2} \left(\frac{y\dot{x}}{x^2} - \frac{\dot{y}}{x} \right) \quad (3b)$$

although the last equation may be a bit intimidating, it will simplify things and will also be very useful later on. Therefore, after careful substitution of Equations 2 into 3, our symmetric oscillator is further simplified in polar coordinates to:

$$\dot{r} = 0 \quad (4a)$$

$$\dot{\theta} = -\omega \quad (4b)$$

The interpretation of these equations is straightforward: the radius stays fixed (no temporal rate of variation) and the angle decrease at a constant rate of ω , i.e. moving clockwise with constant velocity. This is exactly the same as what we started with in the previous part: presenting the simplest possible oscillator on the circle, with the state moving at constant angular velocity. However now we derived it from a physical system with two variables (the harmonic oscillator) after a change of coordinates.

Here are the corresponding definitions in the SuperCollider language (for usage, see example file):

Listing 1: The symmetric oscillator and the oscillation on the circle.

```
// harmonic oscillator
(
Fb1ODEdef(\harmonic, { |t, state, omega = 1.0|
```

¹We choose $\omega = \sqrt{k}$ so that substituting in equation 1, we get the standard differential equations for the harmonic oscillator: $\ddot{v} = -\sqrt{k^2}x = -kx$

```

    var x = state[0];
    var v = state[1];
    [
        { v },
        { omega.squared.neg * x }
    ]
}, t0: 0, y0: [1, 0]); // the initial conditions y0 are the initial conditions of state.
)

// solve the system in sound
(
Ndef(\harmonic, {|omega=0.5|
    var state = Fb1_ODE.ar(\harmonic, [omega], 400, y0:[1,0], leakDC: false);
    state * 0.2
}).play;
)

// harmonic oscillator with constant radius: omega is factored into both equations
(
Fb1_ODEdef(\harmonic, { |t, state, omega = 1.0|
    var x = state[0];
    var y = state[1];
    [
        { omega * y },
        { omega.neg * x }
    ]
}, t0: 0, y0: [1, 0]);
)

// harmonic oscillator with constant radius in polar coordinates
(
Fb1_ODEdef(\harmonic, { |t, state, omega = 1.0|
    var r = state[0]; // radius
    var theta = state[1]; // angle
    [
        { 0 }, // we do not change the radius (amplitude)
        { omega.neg } // the angular frequency is constant: multiply by a constant factor
    ]
}, t0: 0, y0: [1, 1]);
)

// harmonic oscillator with constant radius in polar coordinates
// the first equation was dropped, because it equals zero
(
Fb1_ODEdef(\harmonic, { |t, state, omega = 1.0|
    var theta = state[0]; // angle
    [
        { omega.neg }
    ]
}, t0: 0, y0: [1]);
)

// solve the system in sound
// here the output is converted back in cartesian coordinates
// the sine of theta is our wave form
(
Ndef(\harmonic, {|omega=0.5, gamma = 0.01|
    var state = Fb1_ODE.ar(\harmonic, [omega, gamma], 400, y0:[1,0], leakDC: false);
    state.sin * 0.2
}).play;
)

```

Implementation of computational solutions that grow indefinitely. The final definition of the harmonic oscillator is extremely simple: the only variable (the angle θ) grows linearly (in this case in the negative range), and its projection into cartesian coordinates naturally produces a repeating sine wave form. Since the computational solution requires a numerical representation in computer memory, we run into problems: the variable's floating point representation loses precision as it grows. This is easily solved if we remember the topology of the variable θ : since it is circular, what seemed like an infinite growth is merely an eternal return in a circle. This means that here, many values that are different on a line are identified with each other: for example, returning after a full circle, the

value 2π is the same value as 0, the value 0.4 is the same as $0.4 + 2\pi$, and more generally a value θ is equivalent to all $\theta + 2n\pi$, where $n \in \mathbb{Z}$. In our definition of differential equations, this equivalence can be expressed as a modulo operation, which is applied to the resulting state after each integration step:

Listing 2: Projecting θ into a cyclic domain by a modulo operation.

```
(  
Ndef(\harmonic, {|omega=0.5|  
    var state = Fb1_ODE.ar(\harmonic, [omega].lag, 400, y0:[1], leakDC: false,  
        compose: { |state|  
            var theta = state[0];  
            [theta % 2pi]  
        }  
    );  
    state.sin * 0.2  
}).play;  
)
```

The harmonic oscillator with dissipation. As an additional observation, note that if we now introduce dissipation, which should be added only to the second equation, the symmetry is broken, and the trajectory is no longer on a circle but instead decays into a spiral toward the center as we have seen before.

However, if we move away from the strict physical interpretation, we can introduce dissipation symmetrically in both equations, resulting in the system given by equations (2) transforming into the following:

$$\dot{x} = \omega y - \gamma x \quad (5a)$$

$$\dot{y} = -\omega x - \gamma y \quad (5b)$$

where $\gamma > 0$ is the dissipation coefficient.

In this case, by introducing dissipation symmetrically, the system retains a relatively simple structure, allowing us to analyze how the trajectories change. Applying the transformation to polar coordinates again, we can write:

$$\dot{r} = -\gamma r \quad (6a)$$

$$\dot{\theta} = -\omega \quad (6b)$$

The interpretation of these equations is also straightforward: the radius r decreases exponentially at a rate controlled by γ , while the angle θ continues to decrease at a constant velocity ω . This describes a spiral motion toward the center instead of a perfect circular motion.

Listing 3: The symmetric oscillator and the oscillation on the circle.

```
// harmonic oscillator with dissipation in both variables  
(  
Fb1_ODEDef(\harmonic, { |t, state, omega = 1.0, gamma = 0.01|  
    var x = state[0];  
    var y = state[1];  
    [  
        { (omega * y) - (gamma * x) },  
        { (omega.neg * x) - (gamma * y) }  
    ]  
, t0: 0, y0: [1, 0]);  
)  
  
// harmonic oscillator with dissipation in polar form  
(  
Fb1_ODEDef(\harmonic, { |t, state, omega = 1.0, gamma = 0.01|  
    var r = state[0];  
    var theta = state[1];  
    [  
        { gamma.neg * r },  
        { omega.neg }  
    ]  
, t0: 0, y0: [1, 0]);  
)
```

1.2 Revisiting the self-oscillator. Hopf Normal Form

At this point, we can also consider a way to introduce negative dissipation symmetrically, no longer as a term affecting only the velocity but as one influencing both equations. The simplest way to do this is as follows:

$$\dot{x} = \omega y + \gamma x - x(x^2 + y^2) \quad (7a)$$

$$\dot{y} = -\omega x + \gamma y - y(x^2 + y^2) \quad (7b)$$

Here, the last term may seem complex until we realize that $(x^2 + y^2)$ is simply the square of the radius. In fact, when we switch to polar coordinates, the system becomes:

$$\dot{r} = \gamma r - r^3 \quad (8a)$$

$$\dot{\theta} = -\omega \quad (8b)$$

In these equations, the effect of the negative dissipation on the radius becomes evident. When $\gamma = 0$, we observe a Hopf bifurcation. To understand this, note that the angular dynamics are trivial (a constant rotation), so we focus on the radial dynamics as a one-dimensional system. Trajectories rotate clockwise around the origin, governed by the dynamics of r .

In Figure 1 we illustrate the behavior of \dot{r} for three values of γ , before, on and after the Hopf bifurcation. For negative values of γ (left panel), the functional form of $\gamma r - r^3$ ensures that \dot{r} is negative for all r , causing all trajectories to decay to the origin ($r = 0$). This makes the origin the only fixed point, which is stable (represented as a black dot in the figure). For $\gamma > 0$ (right panel), the functional form $\gamma r - r^3$ is positive for small r (since the γr term dominates over r^3), leading to negative dissipation and trajectories that tend toward a fixed radius, the solution of $\dot{r} = 0$. This stable fixed point is shown as a black dot in the figure, while the origin becomes an unstable fixed point (depicted as a white dot).

The stable solution corresponds to a fixed radius $r = \sqrt{\gamma}$, which is the solution to $\dot{r} = 0$. This implies that all trajectories approach a stable limit cycle with radius $r = \sqrt{\gamma}$, rotating clockwise with angular frequency ω . The intermediate situation ($\gamma = 0$) shown in the central panel of Figure 1 corresponds to the moment of the Hopf bifurcation, where the slope of the function $\gamma r - r^3$ becomes horizontal at the origin, signaling the bifurcation.




Figure 1: The cubic function $f(r) = \gamma r - r^3$ that corresponds to the rate of variation of the radius $r > 0$ (Equation 8.a), for three values of γ . Fixed point appear when $f(r) = 0$ and they are stable (black dots) when the slope of the function in negative and unstable (white dot) when the slope is positive. Crossing $\gamma = 0$ leads to a loss of stability of the fixed point at the origin and the appearance of a stable fixed point at $r = \sqrt{\gamma}$ (Hopf bifurcation).

The full picture of the dynamics of the system governed by equations (7) is displayed on Figure 2 for varying dissipation coefficients γ . For $\gamma < 0$, the system exhibits decaying oscillations with the radius $r(t)$ diminishing and the trajectory spiraling inward. For small positive γ , the radius stabilizes at a finite value, leading to a limit cycle, while for larger γ , the radius stabilizes quickly after a short transient, forming larger limit cycles. This demonstrates the transition from stable equilibrium to stable limit cycle, characteristic of a Hopf bifurcation. Note that the time evolution of the r variable displayed in the bottom row can be interpreted as the envelope of the oscillation of y displayed on the top row.

This system, whether in its cartesian (Equations 7) or polar (Equations 8) form, is known as the Normal Form of the Hopf equation. In dynamical systems, the normal form is a simplified version of a system of differential equations near a critical point, obtained through coordinate transformations, that retains the essential features of the dynamics while discarding higher-order terms. It is a powerful tool to study bifurcations and classify qualitative behavior in nonlinear systems.




Figure 2: Time evolution of the y variable (top) and phase portraits (middle) for the self-oscillator given by Equations 7 and time evolution of the r variable (bottom) for the same system in the polar form given by Equations 8. Three sets of parameters are given: before the Hopf bifurcation with $\gamma = -0.05$ (left), after the Hopf bifurcation with a small $\gamma = 0.1$ value (center) and with a high $\gamma = 1.5$ value. The blue and orange trajectories result from two different initial values.

Listing 4: Exploring the Hopf bifurcation of a self-oscillator.

```

(
Fb1.ODEDef(\harmonic, { |t, state, omega = 1.0, gamma = 0.01|
    var r = state[0];
    var theta = state[1];
    [
        { gamma * r - r.cubed },
        { omega.neg }
    ]
}, t0: 0, y0: [1, 0]);
)

(
Ndef(\harmonic, {|omega=0.5, gamma = 0.01|
    var state = Fb1.ODE.ar(\harmonic, [omega, gamma], 400,
        y0: \initState.kr([1, 0]),
        leakDC: false,
        compose: { |state| var omega = state[0], theta = state[1]; [omega, theta % 2pi] }
    );
    var theta = state[1];
    theta.sin * 0.2
}).play;
)

// restart the system every 2 seconds with different initial states
// slowly moving across the Hopf bifurcation
(
Tdef(\x, {
    20.do { |i|

```

```

        2.wait;
        Ndef(\harmonic).send([
            \omega, 0.5,
            \gamma, i.linlin(0, 20, -0.05, 0.5).postln,
            \initState, [1.0, 1.0].rand,
        ]);
    });
}).play
)

```

1.3 The self modulated oscillator and the first derivation of the Adler's equation

We now explore one possible way to couple oscillators. First, we write the symmetric form of the oscillator without dissipation but incorporate coupling as a modulation on the angular frequency. Specifically, we transform each ω_i of the i -th oscillator into $\omega_i + \lambda_i$, where λ_i is a weighted sum over the x variables of the other oscillators. This means that, in general, the system for the i -th oscillator can be written as:

$$\dot{x}_i = (\omega_i + \lambda_i)y_i \quad (9a)$$

$$\dot{y}_i = -(\omega_i + \lambda_i)x_i \quad (9b)$$

$$\lambda_i = \alpha_{i1}x_1 + \alpha_{i2}x_2 + \dots + \alpha_{ij}x_j + \dots \quad (9c)$$

Here, α_{ij} are coupling coefficients that determine the influence of oscillator j on oscillator i .

To begin with the simplest case, let us consider an oscillator coupled "with itself," meaning that only α_{11} is nonzero. In this case, the previous equations simplify to:

$$\dot{x} = \omega y + \alpha xy \quad (10a)$$

$$\dot{y} = -\omega x - \alpha x^2 \quad (10b)$$

For simplicity, we have removed the subscripts since we are dealing with a single oscillator.

Listing 5: Sonification of a simple self coupled oscillator.

```

(
Fb1.ODEdef(\selfmodulation, { |t, state, alpha = 0, omega = 0.5|
    var x, y;
    x = state[0];
    y = state[1];

    [
        { (omega * y) + (alpha * x * y) },
        { ((omega * x) + (alpha * x * x)).neg },
    ]
}, y0: [1, 1]);

Ndef(\selfmodulation, {
    var l = \lag.kr(0.2);
    var params = [\alpha.kr(0.02, l), \omega.kr(0.5, l)];
    var eq = Fb1.ODE.ar(\selfmodulation, params, 700, y0: [0, 1], leakDC: false);
    0.1 * eq[0];
}).play
)

```

The dynamical behavior of the self-modulated oscillator system is shown in Figure 3 for different values of the parameter α with $\omega = 0.5$.

In the left panel, for $\alpha = 0$, the system behaves as the symmetric oscillator, with $x(t)$ oscillating sinusoidally with constant amplitude and the trajectory in the (x, y) phase space being a perfect circle centered at the origin. There are two initial conditions with amplitudes 0.5 (blue) and 1 (orange) that lead to identical dynamics, differing only in scale.

In the middle panel, for $\alpha = 0.49$, the system exhibits a more interesting behavior. While the trajectories in phase space still describe circles with fixed radii corresponding to the initial amplitudes, the oscillations are no longer uniform. For negative x values, the flux slows down, which is particularly evident for the orange trajectory

with an amplitude of 1. The signal becomes nearly a sequence of pulses, with long periods of slow motion between bursts, resembling the non-uniform oscillations on a circle that we have seen in the first chapter.

In the right panel, for $\alpha = 0.6$, the change in the dynamics become even more evident. The blue trajectory with an initial amplitude of 0.5 now also shows a noticeable slowing down for negative x values. However, the orange trajectory with an amplitude of 1 no longer oscillates but instead ends at a fixed point on the circle. Additionally, a third trajectory, shown in green, starts near the orange trajectory and also ends at the fixed point but approaches it from the opposite direction. This change of behavior indicates that for the dynamics taking place on the circle of radius 1 a Saddle-Node bifurcation occurred.




Figure 3: Time evolution of the x variable (top) and phase portraits (middle) for the self-modulated oscillator given by equations 10 with frequency $\omega = 0.5$ and three values of the parameter α . The different colors correspond to different initial conditions. In the left column the oscillators are uncoupled. In the center column the oscillator that starts with radius one (orange line) is slowed down for negative x values. In the right column two initial conditions that starts with radius one (orange and green lines) converge to an attractor node through different paths after the Saddle-Node bifurcation.

In all cases, the dynamics take place on a circle, so while written as a two-variable system, the temporal dynamics effectively occur in the phase dimension alone. This can be demonstrated by carefully transforming to polar coordinates using the chain rule in Equation 3. The result is:

$$\dot{r} = 0 \quad (11a)$$

$$\dot{\theta} = -\omega - \alpha' \cos \theta \quad (11b)$$

Here, as anticipated, there is no radial dynamics, and the entire system can be described by the second equation, showing that the dynamics is now effectively one-dimensional. In this last equation, α' is the value of α scaled by the fixed radius of the oscillation, which depends on the initial condition. The equation for θ describes the phase dynamics and can be represented graphically as $f(\theta)$ versus θ , as done when we were studying oscillation on the circle.

Then, a Saddle-Node bifurcation occurs when this function touches the horizontal axis, i.e., when $\alpha' = \omega$. For the blue trajectory in Figure 3, with radius $r = 1$, the bifurcation occurs at $\alpha = 0.5$, as observed in the transition from the central to the right panel. For the orange trajectory with a smaller radius, the bifurcation requires $\alpha = 1$ due to the scaling by the radius.

Therefore the dynamics that describe the self-modulated oscillator is effectively one-dimensional and its phase dynamics is governed by the Adler's equation 11.b, undergoing a Saddle-Node bifurcation that ends with the oscillations that happen when $|\alpha'| < \omega$.

Listing 6: Reformulation of the simple self-coupled oscillator in polar coordinates.

```

Fb1_ODEdef(\selfmodulation, { |t, state, alpha = 0, omega = 0.5|
    var theta = state[0];
    [
        { omega.neg - (alpha * cos(theta)) }
    ]
}, y0: [1, 1]);

Ndef(\selfmodulation, {
    var l = \lag.kr(0.2);
    var params = [\alpha.kr(0.02, l), \omega.kr(0.5, l)];
    var eq = Fb1_ODE.ar(\selfmodulation, params, 700, y0: [0, 1], leakDC: false);
    0.1 * eq[0];
}) .play
)

```

Listing 7: The simple self-coupled oscillator in polar coordinates with a remapped α .

```

// remap alpha to a selfMod parameter to express the degree of self modulation independent of omega
(
Ndef(\selfmodulation).clear;
Ndef(\selfmodulation, {
    var l = \lag.kr(0.2);
    var factor = \selfMod.kr(0, 1);
    var omega = \omega.kr(0.5, 1);
    var alpha = omega * factor;
    var params = [alpha, omega];
    var eq = Fb1_ODE.ar(\selfmodulation, params, 700, y0: [0, 1], leakDC: false);
    0.1 * eq[0].sin;
}) .play
)

// display parameters and signals
(
ControlSpec.add(\omega, [0.01, 4, \expl]);
ControlSpec.add(\selfMod, [0, 1.5, \lin]);
Ndef(\selfmodulation).gui;
Ndef(\selfmodulation).scope;
)

```

1.4 Two mutually modulated oscillators

Now, we return to the general form introduced for coupling oscillators via modulation of their angular frequency, given in equations (9). We consider the case of two oscillators mutually modulating each other. The system can be written as:

$$\dot{x}_1 = \omega_1 y_1 + \alpha_{21} x_2 y_1 \quad (12a)$$

$$\dot{y}_1 = -\omega_1 x_1 - \alpha_{21} x_2 x_1 \quad (12b)$$

$$\dot{x}_2 = \omega_2 y_2 + \alpha_{12} x_1 y_2 \quad (12c)$$

$$\dot{y}_2 = -\omega_2 x_2 - \alpha_{12} x_1 x_2 \quad (12d)$$

Here, x_1, y_1 correspond to the variables of the first oscillator with natural frequency ω_1 , and x_2, y_2 correspond to the second oscillator with natural frequency ω_2 . The parameters α_{12} and α_{21} represent the coupling strengths, which are not necessarily equal.

Surprisingly, in this case as well, all the dynamics occur on a circle with a radius equal to the initial amplitude. What is more interesting is that the phase dynamics in this case are even richer than in the previous example. In Figure 4, we display the behavior of the variable x for both oscillators: x_1 is shown in blue, and x_2 in orange. The dynamics in the phase space are not shown because they are trivial, occurring entirely on a circle with radius equal to the initial amplitude of the signal. It is important to note that, unlike the previous figures, here we have a single initial condition, and the colors correspond to the two oscillators. In all cases, the initial condition was $x = 1$ and $y = 0$ for both oscillators.

In the Figure 4, the top row shows a unidirectional coupling where oscillator 2 influences oscillator 1, except for the top-left corner where the oscillators are uncoupled. The bottom row displays the dynamics for bidirectional

coupling. As expected, in the absence of coupling, each oscillator evolves independently according to its own ω , resulting in two separate signals. In the unidirectional case, it is clear how the second oscillator's variable (orange) modulates the first oscillator, altering the waveform in a manner similar to frequency modulation (FM). This demonstrates how FM behavior emerges naturally from these dynamical systems. For bidirectional coupling, the dynamics become more complex, resulting in a richer timbral texture.

Now, let us transform again into polar coordinates. Fortunately, in this case as well, the dynamics simplify: the equations for r_1 and r_2 are null, and the dynamics are governed entirely by the following differential equations for the phases of the oscillators:

$$\dot{\theta}_1 = -\omega_1 - \alpha'_{12} \cos \theta_2 \quad (13a)$$

$$\dot{\theta}_2 = -\omega_2 - \alpha'_{21} \cos \theta_1 \quad (13b)$$

Here, the prime ('') again is a reminder that α is scaled by the radius or amplitude of the oscillator, which depends on the initial condition for system given by Equation 12. But we can easily absorb that amplitude of the oscillator in the definition of the coupling coefficient so that the Equations 13 do not depend on the initial condition.




Figure 4: Time evolution of the x_1 (blue) and x_2 (orange) variables for the two mutually modulated oscillators described by equations 12 for different values of the parameters. The tuples $\alpha = (\alpha_1, \alpha_2)$ and $\omega = (\omega_1, \omega_2)$ are displayed on top. IN the top left panels the oscillators are uncoupled.

This coupling via frequency modulation can now be extended to more oscillators by rewriting the dynamics on the circle. That is, the equation for the phase evolution of the i -th oscillator becomes:

$$\dot{\theta}_i = \omega_i + \sum_j \alpha_{ij} \cos(\theta_j) \quad (14)$$

where the negative signs are incorporated into the parameters. This form is much more efficient for studying oscillators coupled via modulation.

2 Coupled Oscillators on the circle

2.1 General formula for coupled oscillators: the Kuramoto Model

The method of coupling oscillators through mutual modulation is not the only approach, nor is it the most commonly encountered. Perhaps the most general and widely studied framework is the **Kuramoto model**, which couples oscillators via their phase differences. The model is described by the following equation:

$$\dot{\theta}_i = \omega_i + \sum_j \alpha_{ij} \sin(\theta_j - \theta_i) \quad (15)$$

Here, θ_i represents the phase of the i -th oscillator, ω_i is its natural frequency, and α_{ij} is the coupling strength between oscillators i and j . The interaction term, $\sin(\theta_j - \theta_i)$, reflects the dependence of the coupling on the phase difference between the oscillators.

The Kuramoto model was introduced to study synchronization phenomena in systems of coupled oscillators. Unlike coupling methods that directly modulate frequency or amplitude, the Kuramoto model captures how oscillators adjust their phases in response to interactions with others in the network. This makes it particularly suited to describing a wide range of a great variety of systems, among them the famous metronome assembly that being out of phase or with slightly different frequencies synchronize “spontaneously”. Basically what is at work here is a force that is greater the greater the phase difference between the oscillators and becomes zero when the oscillators are in phase, thus forcing them to synchronize even when their natural frequencies are different. This synchronization occurs for a threshold value of the alpha couplings (which is commonly scaled with a K parameter) where a phase transition occurs. For lower values of the parameter before synchronization, a variety of other phenomena occur, among them a chaotic regime.

2.2 Two simple coupled oscillators and the second derivation of the Adler's equation

For the case of two coupled oscillators, the Kuramoto model (assuming symmetric coupling and no self-coupling, i.e. $\alpha_{12} = \alpha_{21} = \alpha$ and $\alpha_{11} = \alpha_{22} = 0$) simplifies to:

$$\dot{\theta}_1 = \omega_1 + \alpha \sin(\theta_2 - \theta_1), \quad (16a)$$

$$\dot{\theta}_2 = \omega_2 + \alpha \sin(\theta_1 - \theta_2). \quad (16b)$$

We can now write the equations in terms of the phase difference between the oscillators $\theta_d = \theta_2 - \theta_1$, we subtract the first equation from the second:

$$\dot{\theta}_d = \dot{\theta}_2 - \dot{\theta}_1 \quad (17a)$$

$$= (\omega_2 - \omega_1) - 2\alpha \sin(\theta_d) \quad (17b)$$

where we use the fact that $\sin(\theta_1 - \theta_2) = -\sin(\theta_2 - \theta_1)$ Defining also the frequency difference between the oscillators as $\omega_d = \omega_2 - \omega_1$, we arrive to the following equation:

$$\dot{\theta}_d = \omega_d - 2\alpha \sin(\theta_d). \quad (18)$$

which is completely analogous to the Adler's equation (the use of *sin* instead of *cos* only shifts the position of the fixed points after the Saddle-Node bifurcation). The Saddle-Node bifurcation now will occurs when $2\alpha = \omega_d$, that means when the coupling strength equals half the frequency difference between the oscillators. This result makes intuitive sense: the closer the natural frequencies of the oscillators are, the smaller the coupling required to achieve synchronization.




Figure 5: Time evolution of the sine of the phases (top row) for the two coupled oscillators of the Kuramoto model described by equations 16, and the sine of the phase difference between them (bottom row) expressed by equation 18 for different values of the coupling parameter α . The frequencies of the oscillators are $\omega_1 = 0.5$ and $\omega_2 = 1.1$. Therefore $\omega_d = 0.6$ and the Saddle-Node bifurcation occurs when $\alpha = 0.3$.

It is now interesting to interpret this system considering that it is an equation for the evolution of the phase difference between two oscillators, or it could also be seen as the detuning. If this system oscillates, it means that the oscillators are not in phase and each one operates with its own frequency. In this case, the phase difference simply grows with time. Conversely, if this system does not oscillate and stays at a fixed point, it means that the oscillators defined by equations 16 are synchronized and remain phase locked. This is why Adler's equation is also known as the *phase lock* equation.

This can be seen in Figure 5. In the upper row, we show the sine of the phases of the two oscillators defined by equations 16, and in the lower row, the evolution of the corresponding phase difference given by the transformed system of equations 18. The frequencies of the oscillators are $\omega_1 = 0.5$ and $\omega_2 = 1.1$. For $\alpha = 0.1$, the oscillators are weakly coupled, and the phase difference oscillates almost sinusoidally, reflecting their detuning. In the central panels, α approaches the critical value for the saddle-node bifurcation, $\alpha = \omega_d/2 = 0.3$. In the signal in the lower panel we can recognize the behavior of the Adler equation as we approach the saddle-node bifurcation: we have long periods with the signal almost horizontal and every now and then a pulse. However, in the context that the variable we are describing is a phase difference this can be interpreted as long periods in which the phase difference remains constant (or latched) and every now and then a slip in which the oscillators decouple and as can be seen in the top center panel "lose a cycle" of the faster oscillator and re-latch. Finally, for $\alpha = 0.6$, the oscillators become phase-locked, and the phase difference rapidly converges to a stable fixed point. In this case, the system reaches strong synchronization, and the phase difference approaches zero as the coupling strength increases further.

2.3 Three uniformly coupled oscillators

We can go back to the original formula of the Kuramoto model and propose a simplification that will serve to better study the system. The idea is that all α_{ij} couplings are either zero or a fixed value K (that letter is a tribute to Kuramoto). This K is be the critical parameter that characterizes the transition from the behavior of unsynchronized oscillators to synchronized oscillators and this transition will happen for a specific value of K globally, i.e. all oscillators will be synchronized when we cross that value. The α_{ij} values will define the topology of the connection between the oscillators. For example, if all the α_{ij} are equal to one we have a topology where all are connected to all. Or for example if the α_{ij} is different from zero only for consecutive oscillator indices (including the last one with the first one) we have a ring topology.

For the case of three oscillators there is no difference between these topologies and having all $\alpha_{ij} = 1$ we can write:

$$\dot{\theta}_1 = \omega_1 + K (\sin(\theta_2 - \theta_1) + \sin(\theta_3 - \theta_1)) \quad (19a)$$

$$\dot{\theta}_2 = \omega_2 + K (\sin(\theta_1 - \theta_2) + \sin(\theta_3 - \theta_2)) \quad (19b)$$

$$\dot{\theta}_3 = \omega_3 + K (\sin(\theta_1 - \theta_3) + \sin(\theta_2 - \theta_3)) \quad (19c)$$




Figure 6: Three mutually coupled oscillators described by equations 19 with parameter values $\omega_2 = 0.47$, $\omega_3 = 0.65$ and three different values of the coupling parameter K displayed on top. The first row displays the time evolution of the sine of the phases that were shifted for clarity (from bottom to top the oscillators from 1 to 3). The second row displays the phase differences between the oscillator pairs 1-2 (blue) 1-3 (red) and 2-3 (green).

The behavior of this system is no longer reducible to a single Adler equation. Although it can be expressed as equations for the relative phase differences between the oscillators in this case also three variables are involved and coupled together. In figure 6 we show in the upper row the behavior of theta variables 1 to 3 for the system described by the equations 19 of the three oscillators with frequencies $\omega_1 = 0.4$, $\omega_2 = 0.47$, $\omega_3 = 0.65$ and three coupling values. The lower row shows the behavior for the relative phase differences between pairs of oscillators.

As the coupling parameter increases, the time evolution of the phase differences becomes sharper with more spaced shots. In the right column for example only a few peaks are seen that adjust the phase every now and then and for larger values complete tuning is achieved and the phase difference between the oscillators goes to a stationary value.

2.4 N-chain of oscillators coupled locally

We could now ask ourselves how this generalizes to the case of many oscillators (let's call N the number of oscillators). Interestingly if the oscillators are all coupled together as in the previous case, i.e. all α equal to 1 then the system is greatly simplified. The result is that the system can be described as a system of N oscillators coupled only through a mean field equal to the sum of the action of all oscillators and even an analytical solution can be found for the $N \rightarrow \infty$ case.

However, here we will be interested in another case of a set of N oscillators coupled only locally between their first neighbors. The simplest example is that of a one-dimensional chain or ring where each oscillator is coupled with only one neighbor on the left and one on the right as shown in the Figure 7. Eventually we can close the chain by coupling the N th oscillator with the first one and then we have a ring.



Figure 7: A chain of N simple oscillators $\theta = \omega$ locally coupled between first neighbours. If the last and first oscillators are also mutually coupled we have a ring topology

The dynamical equation for a generic oscillator in the middle (that is for $1 < i < N$) is:

$$\dot{\theta}_i = \omega_i + K (\sin(\theta_{i-1} - \theta_i) + \sin(\theta_{i+1} - \theta_i)) \quad (20)$$

while for the oscillators at the ends of the chain if we couple them we have:

$$\dot{\theta}_1 = \omega_1 + K (\sin(\theta_N - \theta_1) + \sin(\theta_2 - \theta_1)) \quad (21a)$$

$$\dot{\theta}_N = \omega_N + K (\sin(\theta_{N-1} - \theta_N) + \sin(\theta_1 - \theta_N)) \quad (21b)$$




Figure 8:

We will show the results for a ring of $N = 1000$ oscillators with random values of ω uniformly distributed between 0.1 and 1. Of course it is not practical to display the traces of the time evolution 1000 phases. Therefore we will adopt a phase plot diagram where the horizontal axis corresponds to the oscillator number, the vertical axis to time and the cosine of the phase is displayed as a color ranging from black (when the cosine is equal to -1) to bright yellow (when equals to 1). These diagrams are also called spatiotemporal, because we can imagine the N oscillators arranged in space, so the horizontal axis represents their position in space while the vertical axis shows their temporal evolution. A relevant feature that can emerge in these diagrams is that of organized structures or

spatiotemporal patterns that in many cases are an emergent phenomenon of the system and are very present in many biological systems.

In Figure 8 we display a phase plot for the system given by equations 20 and 21 for four different values of the coupling parameter K . When there is no coupling (top left) the phases are random but as soon as there is some coupling ($K = 0.3$, top right) some structure starts to appear but on the small scale. As the coupling increases a more clear spatiotemporal pattern of ripples like the ones made by the waves in sand appears.

2.5 N-chain of oscillators mutually modulated locally

To show another interesting aspect of coupled systems is the emergence of patterns and structures from the interaction, something that can be seen in Figure 8 but that will be much more evident if instead of phase oscillators we come back to the coupling through modulation that was written in a general form in equation 14.

We will couple N excitable systems in a similar way as we did with the oscillators in a linear chain with the ends connected. In this case the equations are:

$$\dot{\theta}_i = \omega_i + \alpha (\cos(\theta_{i-1}) + \cos(\theta_{i+1})) \quad : \quad 1 < i < N \quad (22a)$$

$$\dot{\theta}_1 = \omega_1 + \alpha (\cos(\theta_N) + \cos(\theta_2)) \quad (22b)$$

$$\dot{\theta}_N = \omega_N + \alpha (\cos(\theta_{N-1}) + \cos(\theta_1)) \quad (22c)$$

where we are assuming that the α_{ij} are only different from zero for consecutive indices, and as before, we will connect the first and last oscillators.




Figure 9:

2.6 N-chain of excitable systems coupled locally

We will use a simplified version of the Adler equation where there is only one parameter ω and the saddle-node bifurcation occurs when ω crosses the value 1:

$$\dot{\theta} = \omega + \cos(\theta). \quad (23)$$

Recall that when there are two fixed points shortly before the bifurcation the system behaves as excitable. That is, if it receives a perturbation above a certain threshold it responds with a complete phase turn before returning to the steady state.

$$\dot{\theta}_i = \omega_i + K(P(\theta_{i-1}) + P(\theta_{i+1})) \quad (24a)$$

where P is some coupling function. We will use:

$$P(\theta) = 1 + \cos(\theta). \quad (25)$$

and, as before we will connect the first and last oscillators.




Figure 10: